N72-20181

NASA TECHNICAL MEMORANDUM

NASA TMX-64640

LINEAR INTERPOLATION OF FOUR-DIMENSIONAL TABULATED DATA FOR COMPUTERS WITH SINGLE SUBSCRIPTED VARIABLE CAPABILITY

By William R. Farr
Preliminary Design Office

November 19, 1971

NASA

George C. Marshall Space Flight Center
Marshall Space Flight Center, Alabama

MSFC - Form 3190 (Rev June 1971)

TECHNICAL REPORT STANDARD TITLE PAGE

1. REPORT NO. TMX-64640	2. GOVERNMENT ACCESSION NO.	3. RECIPIENT'S CATALOG NO.
4. TITLE AND SUBTITLE Linear Interpolation of Four-Dimensional Tabulated Data For Computers with Single Subscripted Variable Capability		5. REPORT DATE November 19. 1971
		6. PERFORMING ORGANIZATION CODE PD-DO-PF
7. AUTHOR(S) William R. Farr		8. PERFORMING ORGANIZATION REPORT #
9. PERFORMING ORGANIZATION NAME AND ADDRESS George C. Marshall Space Flight Center Marshall Space Flight Center, Alabama 35812		10. WORK UNIT NO.
		11. CONTRACT OR GRANT NO.
12. SPONSORING AGENCY NAME AND ADDRESS National Aeronautics and Space Administration Washington, D. C. 20546		13. TYPE OF REPORT & PERIOD COVERED Technical Memorandum
		14. SPONSORING AGENCY CODE

15. SUPPLEMENTARY NOTES

Prepared by Preliminary Design Office, Program Development

16. ABSTRACT

Using only a one-dimensional subscripted variable, a Fortran computer subprogram was developed to linearly interpolate tabulated data of functions of four or less variables, the primary motivation being faster computation.

17. KEY WORDS

linear
interpolation
Fortran
subprogram
tabulated

18. DISTRIBUTION STATEMENT

Unclassified - Unlimited

Erich E. Goerner
Director, Preliminary Design Office

19. SECURITY CLASSIF. (of this report)	20. SECURITY CLASSIF. (of this page)	21. NO. OF PAGES	22. PRICE
Unclassified	Unclassified	24	$3.00

MSFC - Form 3292 (May 1969)

TABLE OF CONTENTS

LIST OF ILLUSTRATIONS

DEFINITION OF SYMBOLS

Symbol	Definition
I, J, K, L	Number of elements in X, Y, Z, W arrays, respectively
II, JJ, KK, LL	Indices of X, Y, Z, W values just less than XS, YS, ZS, WS, respectively
I_1, I_2, I_3, I_4 I'_1, I'_2, I'_3, I'_4	Subscripts of Q for W dimension interpolation
N	Number of elements in one-dimensional Q array
Q	One-dimensional dependent variable array
T_1, T_2, S_1, S_2 T'_1, T'_2, S'_1, S'_2	Values for Z dimension interpolation
U_1, U_2, U'_1, U'_2	Values for Y dimension interpolation
V_1, V_2	Values for X dimension interpolation
X, Y, Z, W	One-dimensional independent variable arrays
XS, YS, ZS, WS	Coordinates of point for which Q is to be evaluated

LINEAR INTERPOLATION OF FOUR-DIMENSIONAL TABULATED DATA FOR COMPUTERS WITH SINGLE SUBSCRIPTED VARIABLE CAPABILITY

INTRODUCTION

At present several numerical methods exist for interpolation of tabulated data of functions of several variables, notable of which are the Gregory-Newton, Gauss, and Lagrangian. These methods depend, however, upon the calculation of finite differences or coefficients to fitted polynomials. If the tabulated data are approximately linear, the above methods are very accurate but the computation time involved is excessive, especially for digital computer applications. Linear interpolation has approximately the same accuracy but with much less computation time.

With the purpose of faster computation, a Fortran digital computer subprogram was developed to linearly interpolate tabulated data of four or fewer dimensions. As a test of the program's proficiency and accuracy, it was used to interpolate approximately linear aerodynamic data which were tabulated as a function of four variables. The results of the linear interpolations were compared with those of Lagrangian interpolations of the same data. The answers varied only in the fourth decimal place; however, the linear routine extracted approximately 1000 values in the time it took the Lagrangian routine to extract 400 values. The slight inaccuracy of the linear method was offset by its inherent speed.

MATHEMATICAL MODEL

The first step in establishing a mathematical model for the linear interpolation of a function Q of the four independent variables X, Y, Z, and W in tabulated form would be to express Q as a subscripted variable of the fourth dimension:

$$Q = Q(X(I),\ Y(J),\ Z(K),\ W(L)) \tag{1}$$

with a single subscripted variable for each of the independent variables, where I, J, K, and L are the number of elements in each array, respectively.

Linear interpolation could then be accomplished easily by finding the location of the desired point Q (XS, YS, ZS, WS) from the independent variable arrays. This method is fine for large computers, but it fails for many smaller machines which allow a maximum of only three subscripts. To permit utilization of the method on any computer with one-dimensional subscripted variable capability, the four-dimensional Q array may be mapped into a one-dimensional array Q^* (N) (the * notation will be dropped for brevity in the following discussion).

The mapping is as follows, selecting a $(3 \times 3 \times 3 \times 3)$ array as an example:

$$
\begin{aligned}
Q(1) &= Q(1, 1, 1, 1) \\
Q(2) &= Q(1, 1, 1, 2) \\
Q(3) &= Q(1, 1, 1, 3) \\
Q(4) &= Q(1, 1, 2, 1) \\
Q(5) &= Q(1, 1, 2, 2) \\
Q(6) &= Q(1, 1, 2, 3) \\
Q(7) &= Q(1, 1, 3, 1) \\
Q(8) &= Q(1, 1, 3, 2) \\
Q(9) &= Q(1, 1, 3, 3) \\
Q(10) &= Q(1, 2, 1, 1) \\
Q(11) &= Q(1, 2, 1, 2) \\
&\vdots \\
Q(N') &= Q(I', J', K', L')
\end{aligned}
\tag{2}
$$

$$N' = (I' - 1) \cdot J \cdot K \cdot L + (J' - 1) \cdot K \cdot L + (K' - 1) \cdot L + L' \ . \tag{3}$$

For the $(3 \times 3 \times 3 \times 3)$ example there will be $I \cdot J \cdot K \cdot L$ or 81 elements in the Q(N) array. Figure 1 depicts the complete mapping for the $(3 \times 3 \times 3 \times 3)$ example. However, because of the form of certain mapping equations [see equations (5)], the number of elements in Q must be increased by $L \cdot (K + 1) + 1$. These added terms are for working storage, and their values must be initially zero. Therefore, the number of elements in Q must be

$$N = I \cdot J \cdot K \cdot L + L \cdot (K + 1) + 1. \tag{4}$$

```
Q[ 1] = Q[1,1,1,1]          Q[28] = Q[2,1,1,1]          Q[55] = Q[3,1,1,1]
Q[ 2] = Q[1,1,1,2]          Q[29] = Q[2,1,1,2]          Q[56] = Q[3,1,1,2]
Q[ 3] = Q[1,1,1,3]          Q[30] = Q[2,1,1,3]          Q[57] = Q[3,1,1,3]
Q[ 4] = Q[1,1,2,1]          Q[31] = Q[2,1,2,1]          Q[58] = Q[3,1,2,1]
Q[ 5] = Q[1,1,2,2]          Q[32] = Q[2,1,2,2]          Q[59] = Q[3,1,2,2]
Q[ 6] = Q[1,1,2,3]          Q[33] = Q[2,1,2,3]          Q[60] = Q[3,1,2,3]
Q[ 7] = Q[1,1,3,1]          Q[34] = Q[2,1,3,1]          Q[61] = Q[3,1,3,1]
Q[ 8] = Q[1,1,3,2]          Q[35] = Q[2,1,3,2]          Q[62] = Q[3,1,3,2]
Q[ 9] = Q[1,1,3,3]          Q[36] = Q[2,1,3,3]          Q[63] = Q[3,1,3,3]
Q[10] = Q[1,2,1,1] I1       Q[37] = Q[2,2,1,1] I1'      Q[64] = Q[3,2,1,1]
Q[11] = Q[1,2,1,2]    T1    Q[38] = Q[2,2,1,2]    T1'   Q[65] = Q[3,2,1,2]
Q[12] = Q[1,2,1,3]          Q[39] = Q[2,2,1,3]          Q[66] = Q[3,2,1,3]
Q[13] = Q[1,2,2,1] I2  U1   Q[40] = Q[2,2,2,1] I2' U1'  Q[67] = Q[3,2,2,1]
Q[14] = Q[1,2,2,2]          Q[41] = Q[2,2,2,2]          Q[68] = Q[3,2,2,2]
Q[15] = Q[1,2,2,3]    T2    Q[42] = Q[2,2,2,3]    T2'   Q[69] = Q[3,2,2,3]
Q[16] = Q[1,2,3,1]          Q[43] = Q[2,2,3,1]          Q[70] = Q[3,2,3,1]
Q[17] = Q[1,2,3,2]    V1    Q[44] = Q[2,2,3,2]    V2    Q[71] = Q[3,2,3,2]
Q[18] = Q[1,2,3,3]          Q[45] = Q[2,2,3,3]          Q[72] = Q[3,2,3,3]
Q[19] = Q[1,3,1,1] I3       Q[46] = Q[2,3,1,1] I3'      Q[73] = Q[3,3,1,1]
Q[20] = Q[1,3,1,2]    S1    Q[47] = Q[2,3,1,2]    S1'   Q[74] = Q[3,3,1,2]
Q[21] = Q[1,3,1,3]          Q[48] = Q[2,3,1,3]          Q[75] = Q[3,3,1,3]
Q[22] = Q[1,3,2,1] I4       Q[49] = Q[2,3,2,1] I4'      Q[76] = Q[3,3,2,1]
Q[23] = Q[1,3,2,2]    U2    Q[50] = Q[2,3,2,2]    U2'   Q[77] = Q[3,3,2,2]
Q[24] = Q[1,3,2,3]    S2    Q[51] = Q[2,3,2,3]    S2'   Q[78] = Q[3,3,2,3]
Q[25] = Q[1,3,3,1]          Q[52] = Q[2,3,3,1]          Q[79] = Q[3,3,3,1]
Q[26] = Q[1,3,3,2]          Q[53] = Q[2,3,3,2]          Q[80] = Q[3,3,3,2]
Q[27] = Q[1,3,3,3]          Q[54] = Q[2,3,3,3]          Q[81] = Q[3,3,3,3]
```

Figure 1. Four-dimensional $(3 \times 3 \times 3 \times 3)$ array mapped into a one-dimensional array.

Associated with the one-dimensional Q array will still be a one-dimensional array for each of the independent variables X, Y, Z, and W. These arrays contain the data points along the respective independent variable ranges, e.g., X may vary from 10 to 100 in increments of 10, thus giving 10 values for the X array. The elements of these arrays must be in nondecreasing order.

METHOD OF LINEAR INTERPOLATION FOR FOUR DIMENSIONS

After the dependent and independent variable arrays have been established, the dependent variable Q may be determined for a given point with independent variables (XS, YS, ZS, WS) by linear interpolation.

Initially, the location of the point's independent variable coordinates within the respective independent variable arrays must be ascertained. For example, XS obviously lies between some X(II) and X(II + 1); also YS lies between some Y(JJ) and Y(JJ + 1), etc. The indices of prime interest are II, JJ, KK, and LL. If any of the point's coordinates happen to coincide with the last entry in the respective array, that corresponding index is equated to the array dimension, e.g., II = I. Also, if any of the point's coordinates lie outside the range of their respective arrays, these variables are assumed to be the endpoints.

Again, referring to the $(3 \times 3 \times 3 \times 3)$ array in Figure 1, the linear interpolation method may be illustrated with sample values for (II, JJ, KK, LL) of (1, 2, 1, 2). These known indices facilitate the determination of the appropriate values of Q, which must be interpolated in each of the four dimensions. As noted in Figure 1, these particular elements of the Q array are subscripted I_1, I_2, I_3, I_4, I'_1, I'_2, I'_3, I'_4, and are obtained from the following mapping equations:

$$
\begin{aligned}
I_1 &= (II - 1) \cdot J \cdot K \cdot L + (JJ - 1) \cdot K \cdot L + (KK - 1) \cdot L + LL \\
I_2 &= I_1 + L \\
I_3 &= I_1 + K \cdot L \\
I_4 &= I_3 + L
\end{aligned}
\tag{5}
$$

The primed subscripts are obtained by adding $J \cdot K \cdot L$ to the preceding unprimed subscripts, respectively. The equation for I_4 forces the Q storage requirement increase referred to above.

A schematic of the necessary linear interpolations in each of the four dimensions is shown in Figure 2. Interpolation of the aforementioned subscripted elements of Q in the W dimension yields the values T_1, T_2, S_1, T'_1, T'_2, S'_1, S'_2. These values are then interpolated in the Z dimension to produce U_1, U_2, U'_1, U'_2, which are interpolated to obtain V_1 and V_2 in the Y dimension. Q(XS, YS, ZS, WS) is then obtained from the X dimension interpolation of V_1 and V_2. The linear interpolations are of the form

$$Q = V_1 + \frac{XS - X(II)}{X(II + 1) - X(II)} \cdot V_2 \quad . \tag{6}$$

The required value of Q is thus easily obtained once the appropriate Q subscripts are known.

METHOD OF LINEAR INTERPOLATION IN FEWER THAN FOUR DIMENSIONS

The previous equations pertaining to linear interpolation of four-dimensional tabulated data may also be used with 1, 2, or 3 dimensions. The only necessary requirements for three dimensions are that L and LL be equated to one, W(1) be equated to zero, and W(2) be equated to a positive real number.

In two dimensions the three-dimensional requirements must be met in addition to similar requirements on the Z dimension parameters: $K = 1$, $KK = 1$, $Z(1) = 0$, and $Z(2) > 0$, with additional simplifications:

$$\begin{aligned} I_3 &= I_1 \\ I_4 &= I_2 \quad . \end{aligned} \tag{7}$$

For one dimension the two-dimensional requirements must be met plus similar requirements on the Y parameters: $J = 1$, $JJ = 1$, $Y(1) = 0$, $Y(2) > 0$. There is no need now for the primed indices on Q since the interpolation equations need hold for only the unprimed indices.



Figure 2. Schematic of four-dimensional interpolation.

COMPUTER SUBPROGRAM

Two Fortran computer subprograms were written for linear interpolation of tabulated data of four or fewer dimensions. Listings of these routines are given in Appendix A and B. The routine referred to in Appendix A requires no more than single-subscript capability, whereas that in Appendix B requires double-subscript capability. The computer word storage allocations for the respective routines are Routine A, 727, and Routine B, 718. Routine B is preferred because of its smaller storage allocation and slightly faster computation time. However, it is somewhat more complicated in the expression of its arguments.

The included comment cards at the beginning of each routine should be sufficient to enable easy use of the methods based on the mathematical model presented herein.

APPENDIX A

FOUR-DIMENSIONAL LINEAR INTERPOLATION SUBPROGRAM FOR COMPUTERS WITH SINGLE SUBSCRIPTED VARIABLE CAPABILITY

```
      FUNCTION TLU4D (N,J,K,L,X,Y,Z,W,Q,XS,YS,ZS,WS)
      DIMENSION X(1), Y(1), Z(1), W(1), Q(1), V(2)
C
C
C        FOUR-DIMENSIONAL TABLE LOOK-UP ROUTINE
C
C        Q IS A FUNCTION OF THE FOUR VARIABLES X,Y,Z, AND W.  IN THE
C        TABLE, A ONE DIMENSIONAL ARRAY IS ENTERED FOR EACH VARIABLE AND
C        Q.  THE DIMENSIONS OF THE ARRAYS ARE SPECIFIED AS...
C
C                   N   NUMBER OF ELEMENTS IN  X  ARRAY
C                   J   NUMBER OF ELEMENTS IN  Y  ARRAY
C                   K   NUMBER OF ELEMENTS IN  Z  ARRAY
C                   L   NUMBER OF ELEMENTS IN  W  ARRAY
C
C        THE VALUES OF Q IN THE TABLE ENTRY ARRAY ARE EXPRESSED AS IN
C        THIS EXAMPLE OF A 3,3,3,3 ARRAY...
C
C                                Q(1)  =  Q(1,1,1,1)
C                                Q(2)  =  Q(1,1,1,2)
C                                Q(3)  =  Q(1,1,1,3)
C                                Q(4)  =  Q(1,1,2,1)
C                                Q(5)  =  Q(1,1,2,2)
C                                Q(6)  =  Q(1,1,2,3)
C                                Q(7)  =  Q(1,1,3,1)
C                                Q(8)  =  Q(1,1,3,2)
C                                Q(9)  =  Q(1,1,3,3)
C                                Q(10) =  Q(1,2,1,1)     AND SO ON.
C
C        THE DIMENSION OF Q MUST BE  N*J*K*L + L*(K+1)+1  WHERE THE
C        L*(K+1)+1  TERM IS FOR WORKING STORAGE.  THE ELEMENTS OF THE
C        WORKING STORAGE AREA OF THE ARRAY MUST BE INITIALLY ZEROED OUT.
C
C        GIVEN VALUES FOR THE FOUR VARIABLES (XS,YS,ZS,WS) THE ROUTINE
C        LINEARLY INTERPOLATES FOR THE VALUE OF Q(XS,YS,ZS,WS), THE
C        IF ANY OF THE FOUR VARIABLES XS,YS,ZS,WS LIE OUTSIDE THE
C        RANGE OF THEIR RESPECTIVE ARRAYS, THESE VARIABLES ARE ASSUMED
C        TO BE THE ENDPOINTS,
C        ANSWER BEING EXPRESSED AS TLU4D.
C
C        THE ROUTINE IS WRITTEN FOR FOUR DIMENSIONAL USAGE, BUT MAY BE
C        USED FOR 1,2,AND 3 DIMENSIONS.  FOR A THREE-DIMENSIONAL TABLE
C        L MUST BE SET EQUAL TO 1 AND A DUMMY VARIABLE D(K) MUST BE SET
C        FOR THE W ARRAY AS  D(1)=0.0 AND  D(2)= ANY POSITIVE VALUE
C        GREATER THAN 0.  FOR A TWO-DIMENSIONAL TABLE K AND L MUST BE
C        EQUAL TO 1 WHILE THE DUMMY ARRAY REPLACES BOTH THE W AND Z
C        ARRAYS.   FOR A ONE-DIMENSIONAL TABLE J,K,L MUST ALL BE SET
C        EQUAL TO 1 AND THE DUMMY ARRARY REPLACES THE Y,Z, AND W ARRAYS.
C
C
```

```
      I=1
   10 IF (X(I)-XS) 20,20,50
   20 IF (I-N) 30,40,40
   30 I=I+1
      GO TO 10
   40 II=N
      GO TO 60
   50 II=I-1
   60 IF (J-1) 70,70,80
   70 JJ=1
      KK=1
      LL=1
      GO TO 300
C
   80 I=1
   90 IF (Y(I)-YS) 100,100,130
  100 IF (I-J) 110,120,120
  110 I=I+1
      GO TO 90
  120 JJ=J
      GO TO 140
  130 JJ=I-1
  140 IF (K-1) 150,150,160
  150 KK=1
      LL=1
      GO TO 300
C
  160 I=1
  170 IF (Z(I)-ZS) 180,180,210
  180 IF (I-K) 190,200,200
  190 I=I+1
      GO TO 170
  200 KK=K
      GO TO 220
  210 KK=I-1
  220 IF (L-1) 230,230,240
  230 LL=1
      GO TO 300
C
  240 I=1
  250 IF (W(I)-WS) 260,260,290
  260 IF (I-L) 270,280,280
  270 I=I+1
      GO TO 250
  280 LL=L
      GO TO 300
  290 LL=I-1
C
  300 NN=J*K*L
      IF (II) 320,310,320
  310 II=1
  320 IF (JJ) 340,330,340
  330 JJ=1
  340 IF (KK) 360,350,360
  350 KK=1
```

```
  360 IF (LL) 380,370,380
  370 LL=1
  380 I1=(II-1)*NN+(JJ-1)*K*L+(KK-1)*L+LL
      I2=I1+L
      IF (K-1) 390,390,400
  390 I3=I1
      I4=I2
      GO TO 410
  400 I3=I1+K*L
      I4=I3+L
C
  410 DO 440 I=1,2,1
      W1=(WS-W(LL))/(W(LL+1)-W(LL))
      T1=Q(I1)+W1*(Q(I1+1)-Q(I1))
      T2=Q(I2)+W1*(Q(I2+1)-Q(I2))
      S1=Q(I3)+W1*(Q(I3+1)-Q(I3))
      S2=Q(I4)+W1*(Q(I4+1)-Q(I4))
      U1=T1+(ZS-Z(KK))*(T2-T1)/(Z(KK+1)-Z(KK))
      U2=S2+(ZS-Z(KK))*(S2-S1)/(Z(KK+1)-Z(KK))
      V(I)=U1+(YS-Y(JJ))*(U2-U1)/(Y(JJ+1)-Y(JJ))
      IF (J-1) 420,420,430
  420 V(2)=U2
      GO TO 450
  430 I1=I1+NN
      I2=I2+NN
      I3=I3+NN
      I4=I4+NN
  440 CONTINUE
  450 TLU4D=V(1)+(XS-X(II))*(V(2)-V(1))/(X(II+1)-X(II))
      RETURN
      END
```

APPENDIX B

FOUR-DIMENSIONAL LINEAR INTERPOLATION SUBPROGRAM FOR COMPUTERS WITH DOUBLE SUBSCRIPTED VARIABLE CAPABILITY

```
      FUNCTION TLU4D (M,X,Q,XS)
      DIMENSION Q(1), V(2), P(5), XS(1), M(1), L(4), X(1,1)
C
C
C        FOUR-DIMENSIONAL TABLE LOOK-UP ROUTINE
C
C
C        Q IS A FUNCTION OF THE FOUR VARIABLES X,Y,Z,W AND M IS A ONE-
C        DIMENSIONAL ARRAY WITH FOUR ELEMENTS WHICH ARE THE NUMBER OF
C        ELEMENTS IN THE INDEPENDENT VARIABLE ARRAYS ...
C
C                  N = M(1) = NUMBER OF ELEMENTS IN X ARRAY
C                  J = M(2) = NUMBER OF ELEMENTS IN Y ARRAY
C                  K = M(3) = NUMBER OF ELEMENTS IN Z ARRAY
C                  L = M(4) = NUMBER OF ELEMENTS IN W ARRAY
C
C        WHERE ...,
C                  X(1,N) CORRESPONDS TO THE X ARRAY
C                  X(2,N) CORRESPONDS TO THE Y ARRAY
C                  X(3,N) CORRESPONDS TO THE Z ARRAY
C                  X(4,N) CORRESPONDS TO THE W ARRAY
C
C        AND ALSO ...
C                  XS(1) CORRESPONDS TO THE POINT COORDINATE XS
C                  XS(2) CORRESPONDS TO THE POINT COORDINATE YS
C                  XS(3) CORRESPONDS TO THE POINT COORDINATE ZS
C                  XS(4) CORRESPONDS TO THE POINT COORDINATE WS
C
C
C        THE DIMENSION OF Q MUST BE  N*J*K*L + L*(K+1)+1  WHERE THE
C        L*(K+1)+1  TERM IS FOR WORKING STORAGE.  THE ELEMENTS OF THE
C        WORKING STORAGE AREA OF THE ARRAY MUST BE INITIALLY ZEROED OUT.
C
C        GIVEN VALUES FOR THE FOUR VARIABLES (XS,YS,ZS,WS) THE ROUTINE
C        LINEARLY INTERPOLATES FOR THE VALUE OF Q(XS,YS,ZS,WS), THE
C        ANSWER BEING EXPRESSED AS TLU4D.
C        IF ANY OF THE FOUR VARIABLES XS,YS,ZS,WS LIE OUTSIDE THE
C        RANGE OF THEIR RESPECTIVE ARRAYS, THESE VARIABLES ARE ASSUMED
C        TO BE THE ENDPOINTS.
C
C        THE ROUTINE IS WRITTEN FOR FOUR DIMENSIONAL USAGE, BUT MAY BE
C        USED FOR 1,2,AND 3 DIMENSIONS.  FOR A THREE-DIMENSIONAL TABLE
C        L MUST BE SET EQUAL TO 1 AND A DUMMY VARIABLE D(K) MUST BE SET
C        FOR THE W ARRAY AS  D(1)=0.0 AND  D(2)= ANY POSITIVE VALUE
C        GREATER THAN 0.  FOR A TWO-DIMENSIONAL TABLE K AND L MUST BE
C        EQUAL TO 1 WHILE THE DUMMY ARRAY REPLACES BOTH THE W AND Z
C        ARRAYS.   FOR A ONE-DIMENSIONAL TABLE J,K,L MUST ALL BE SET
C        EQUAL TO 1 AND THE DUMMY ARRARY REPLACES THE Y,Z, AND W ARRAYS.
C
C
```

```
      DO 100 J=1,4,1
      I=M(J)
      DO 10 K=1,I,1
   10 P(K)=X(J,K)
      PS=XS(J)
      IF (M(J)-1) 80,80,20
   20 I=1
   30 IF (P(I)-PS) 40,40,70
   40 IF (I-M(J)) 50,60,60
   50 I=I+1
      GO TO 30
   60 L(J)=M(J)
      GO TO 100
   70 L(J)=I-1
      GO TO 100
   80 DO 90 K=J,4,1
   90 L(K)=1
      GO TO 130
  100 CONTINUE
      DO 120 K=1,4,1
      IF (L(K)) 120,110,120
  110 L(K)=1
  120 CONTINUE
  130 N=M(2)*M(3)*M(4)
      I1=(L(1)-1)*N+(L(2)-1)*M(3)*M(4)+(L(3)-1)*M(4)+L(4)
      I2=I1+M(4)
      IF (M(3)-1) 140,140,150
  140 I3=I1
      I4=I2
      GO TO 160
  150 I3=I1+M(3)*M(4)
      I4=I3+M(4)
  160 DO 190 K=1,2,1
      II=L(1)
      JJ=L(2)
      KK=L(3)
      LL=L(4)
      W1=(XS(4)-X(4,LL))/(X(4,LL+1)-X(4,LL))
      T1=Q(I1)+W1*(Q(I1+1)-Q(I1))
      T2=Q(I2)+W1*(Q(I2+1)-Q(I2))
      S1=Q(I3)+W1*(Q(I3+1)-Q(I3))
      S2=Q(I4)+W1*(Q(I4+1)-Q(I4))
      U1=T1+(XS(3)-X(3,KK))*(T2-T1)/(X(3,KK+1)-X(3,KK))
      U2=S2+(XS(3)-X(3,KK))*(S2-S1)/(X(3,KK+1)-X(3,KK))
      V(K)=U1+(XS(2)-X(2,JJ))*(U2-U1)/(X(2,JJ+1)-X(2,JJ))
      IF (M(2)-1) 170,170,180
  170 V(2)=U2
      GO TO 200
  180 I1=I1+N
      I2=I2+N
      I3=I3+N
      I4=I4+N
  190 CONTINUE
  200 TLU4D=V(1)+(XS(1)-X(1,II))*(V(2)-V(1))/(X(1,II+1)-X(1,II))
      RETURN
      END
```

APPROVAL

LINEAR INTERPOLATION OF FOUR-DIMENSIONAL TABULATED DATA FOR COMPUTERS WITH SINGLE SUBSCRIPTED VARIABLE CAPABILITY

By William R. Farr

The information in this report has been reviewed for security classification. Review of any information concerning Department of Defense or Atomic Energy Commission programs has been made by the MSFC Security Classification Officer. This report, in its entirety, has been determined to be unclassified.

This document has also been reviewed and approved for technical accuracy.

ROBERT M. CROFT
Acting Chief, Flight Mechanics Branch

WILLIAM D. GOLDSBY, JR.
Acting Chief, Performance and Flight Mechanics Division

ERICH E. GOERNER
Director, Preliminary Design Office

JAMES T. MURPHY
Acting Director, Program Development

MSFC—RSA, Ala

N72-20181

DISTRIBUTION

INTERNAL

DIR
Dr. Rees

DEP-T
Dr. Lucas

AD-S
Dr. Stuhlinger

PD-DIR
Mr. Murphy
Mr. Downey
Dr. Mrazek

PD-DO-DIR
Mr. Goerner
Dr. Thomason
Mr. Heyer
Mrs. Andrews

PD-DO-M
Mr. Laue

PD-DO-E
Mr. Digesu

PD-DO-S
Mr. Marshall

PD-DO-P
Mr. Goldsby

PD-DO-PA
Mr. Allen
Mr. Burton
Mr. Zimmerman
Mr. Jump
Mr. Fults
Mr. Sullivan
Mr. Sparks

PD-DO-PM
Mr. French
Mr. Wheeler
Mr. Leonard
Mr. Bradford
Mr. Young
Mr. Gold

PD-DO-PF
Mr. Croft
Mr. Solmon
Mr. Perkinson
Mr. Sumrall
Mr. Lowery
Mr. Farr (10)

S&E-AERO-DIR
Dr. Geissler
Mr. Horn
Mr. Bean

S&E-AERO-G
Mr. Baker

S&E-AERO-A
Mr. Dahm

S&E-AERO-M
Mr. Lindberg

S&E-AERO-D
Dr. Lovingood

A&TS-MS-IP (2)

A&TS-MS-IL (8)

A&TS-MS-H

A&TS-TU (6)

A&TS-PAT

PM-PR-M

EXTERNAL

NASA Headquarters
Washington, D.C. 20546
Attn: Library (5)

NASA Ames Research Center
Mountain View, Calif. 94035
Attn: Library (5)

NASA Langley Research Center
Hampton, Virginia 23365
Attn: Library (5)

NASA Lewis Research Center (5)
Tech. Info. Div.
21000 Brookpark Road
Cleveland, Ohio 44135

NASA Manned Spacecraft Center
Houston, Texas 77058
Attn: Library (5)

Northrop Space Laboratories
Technology Drive
Huntsville, Alabama 35805
Attn: Library (5)

Lockheed Missiles and Space Company
Technology Drive
Huntsville, Alabama 35805
Attn: Library (5)

Scientific and Technical Information Facility (25)
P.O. Box 33
College Park, Maryland 20740
Attn: NASA Representative (S-AK/RKT)